In[ ]:=
```
SetOptions[SelectedNotebook[],
 PrintingStyleEnvironment → "Printout", ShowSyntaxStyles → True]
```

In[ ]:=
```
<< Notation`
```

In[ ]:=
```
(* Simplify the notation for the
 formulas: index vectors based on a subscript notation *)
Notation[ x_n_ ⟺ x_[[n_]] ]
```

In[ ]:=
```
χ[a_?NumericQ] := Exp[-a r^2] Sqrt[a/π];
α = {0.298073, 1.242567, 5.782948, 38.474970};
```

In[ ]:=
```
Plot[Array[χ[α_#] &, {4}] // Evaluate,
 {r, 0, 3}, PlotLegends → "Expressions", PlotRange → All]
```

Out[ ]=



In[ ]:=
```
S = Array[(π/(α_#1 + α_#2))^(3/2) &, {4, 4}];
#[S] & /@ {SymmetricMatrixQ, ArrayPlot, MatrixForm}
```

Out[ ]=

{True,  ,

$$\begin{pmatrix} 12.0975 & 2.91188 & 0.37133 & 0.0230638 \\ 2.91188 & 1.42135 & 0.299025 & 0.022246 \\ 0.37133 & 0.299025 & 0.141565 & 0.018912 \\ 0.0230638 & 0.022246 & 0.018912 & 0.00824921 \end{pmatrix}$$

}

In[ ]:=
```
Q = Array[(2 π^(5/2))/((α_#1 + α_#2) (α_#3 + α_#4) Sqrt[α_#1 + α_#2 + α_#3 + α_#4]) &, {4, 4, 4, 4}];
{SymmetricMatrixQ[Q],
 Grid@Array[ArrayPlot[Q_#1,#2, ImageSize → Tiny] &, {4, 4}], MatrixForm@Q}
```

Out[ ]=

{False, ,

```
90.1588   26
26.0598   13
3.7349    3.
0.241237  0.:
26.0598   8.:
8.39724   4.!
1.3527    1.:
0.092246  0.0!
3.7349    1.
1.3527    0.7
0.271299  0.:
0.0221563 0.0
0.241237  0.0
0.092246  0.0!
0.0221563 0.0:
0.0026428 0.00
```

In[ ]:=

$$H = \text{Array}\left[\frac{3\,\pi^{3/2}\,\alpha_{\#1}\,\alpha_{\#2}}{(\alpha_{\#1}+\alpha_{\#2})^{5/2}} - \frac{4\,\pi}{\alpha_{\#1}+\alpha_{\#2}}\ \&,\ \{4, 4\}\right];$$

```
#[H] & /@ {SymmetricMatrixQ, ArrayPlot, MatrixForm}
```

Out[ ]=

{True, , $\begin{pmatrix} -15.6704 & -6.05651 & -1.75072 & -0.303635 \\ -6.05651 & -2.40744 & -0.871147 & -0.236062 \\ -1.75072 & -0.871147 & 0.141492 & 0.00129568 \\ -0.303635 & -0.236062 & 0.00129568 & 0.312776 \end{pmatrix}$ }

In[ ]:=

$$\text{matNorm}[v\_, M\_] := \frac{v}{\text{Sqrt}[v.M.v]}$$

In[ ]:=

```
oldC = matNorm[ConstantArray[2., {4}], S]
oldC.S.oldC
```

Out[ ]=

```
{0.218418, 0.218418, 0.218418, 0.218418}
```

Out[ ]=

```
1.
```

In[ ]:=

```
diffNormList = {}; (* List of the norm of the differences *)
```

In[ ]:=
```
prec = 10^-6; (* Convergence precision *)
newC = oldC + prec;

While[Norm[oldC - newC] > prec,
 AppendTo[diffNormList, Norm[oldC - newC]];
 (* Append the norm of the difference to the diffNormList list *)
 oldC = newC; (* Overwrite the previous eigenvalues *)
 F = H + Q.oldC.oldC; (* Calculate the matrix F each iteration *)
 {evals, efns} = Eigensystem[{F, S}];
 {evals, efns} = {evals[[#]], efns[[#]]} &@Ordering[evals];
 (* Order the eigenvalues always in the same way for correctness *)
 newC = matNorm[First@efns, S];
 (* Normalize the minimal eigenvector w.r.t. the overlap matrix S *)
 If[newC.S.newC ≠ 1, Abort[], Nothing];
 (* Perform a numerical divergence check *)
]
```

In[ ]:=
```
groundEn = Quantity[2 H.oldC.oldC + Q.oldC.oldC.oldC.oldC, "HartreeEnergy"]
(* Ground energy in Hartree energies *)
```

$$\text{Abs}\left[\frac{\text{groundEn} - -2.903\,E_h}{-2.903\,E_h}\right] \text{// PercentForm}$$

Out[ ]=

$-2.85516\,E_h$

Out[ ]//PercentForm=

1.648%

In[ ]:=
```
fns = #.Array[χ[α#] &, {4}] & /@ efns;
Plot[fns, {r, 0, 3}, PlotRange → Full, PlotLegends → Placed["Expressions", Below]]
```

Out[ ]=



$1.42119\, e^{-38.475 r^2} + 0.866291\, e^{-5.78295 r^2} + 0.384959\, e^{-1.24257 r^2} + 0.0707194\, e^{-0.298073 r^2}$

$-0.96176\, e^{-38.475 r^2} - 0.296127\, e^{-5.78295 r^2} - 0.54995\, e^{-1.24257 r^2} + 0.103159\, e^{-0.298073 r^2}$

$0.104244\, e^{-38.475 r^2} + 1.28628\, e^{-5.78295 r^2} - 0.197489\, e^{-1.24257 r^2} + 0.0126622\, e^{-0.298073 r^2}$

$3.43228\, e^{-38.475 r^2} - 0.261344\, e^{-5.78295 r^2} + 0.0195436\, e^{-1.24257 r^2} - 0.00109799\, e^{-0.298073 r^2}$

In[∘]:=
```
Plot[r Abs[First@fns]^2, {r, 0, 3}, PlotRange → Full]
```
Out[∘]=



In[∘]:=
```
ListStepPlot[Drop[diffNormList, 1], PlotRange → Full]
(* Measurements give 0.05 seconds for the convergence time for an SCF loop *)
```
Out[∘]=